



A Brief Introduction to MPLUS

Mplus is accessible in the following three ways: (1) license a copy from <http://www.statmodel.com/> for your own computer (2) access it at the ODUM Institute and (3) download the free student version from the website cited above if the model of interest is small enough.

Launching Mplus



When Mplus is launched, the window which appears is the input window where syntax is input.

These inputs are saved into .inp files before submitting it for analyses when  is hit.

Once Mplus has finished processing the input, an output window will show up displaying the syntax and model results. A copy of the output file will be saved in the same directory as your input file, with the same name but .out as its extension

To make changes to the model, changes should be made to the input file.

Reading Data and Outputting Sample Statistics

The following are the most commonly used Mplus commands used are:

`title` - Specifies a title that Mplus will print on every page of the output file.

`data` - Specifies where Mplus will locate the data, the format of the data and the names of variables.

`variable` - Defines variable names and can be used in conjunction with `usevariables` to select a subset of variables for analysis.

`analysis` - Defines what type of analysis to perform.

`output` - Specifies what output to be included in the output file.

The `data` and `variable` commands are required for all analysis. All commands must begin on a new line and must be followed by a colon (:). Semicolons (;) separate command options and there can be more than one option per line. The syntax in the input can be no longer than 80 columns.

There are several ways to read in data:

- (1) Raw data
- (2) Triangle Correlation matrix

Note that Mplus can only read text files of the following formats: space-delimited, tab-delimited and comma-delimited.

The data files and input files should be saved in the same directory.

Raw data:

```
title: read in mardia raw data !title of analysis
data:
  file is mardia.dat; !name of file in same directory as input file
  format is free;
variable:
  names are mech vec algebra analysis stats; !variable names
  usevariables mech vec algebra analysis stats; !variables used
analysis:
  type = basic; !means and covariances reported
```

Note that the variable names cannot be longer than 8 characters.

Triangle Correlation matrix:

```
title: read in mardia correlation !title of analysis
data:
  file is mardia_corr2.txt;
  nobservations = 88; !the number of observations must be included
  type=correlation; !type=covariance tells mplus that the data are covariances
variable:
  names are mech vec algebra analysis stats; !variable names
  usevariables mech vec algebra analysis stats; !variables used
analysis:
  type = efa 1 2;
```

Note that `type=basic` cannot be conducted on correlations. In addition, Mplus will *not* read in a full square symmetric correlation matrix. If a covariance matrix is read in by using `type=covariance`, Mplus will find it incompatible with running an analysis of `type=efa`.

Doing a Factor Analysis

```
title: mardia; !title of analysis

data:
  file is mardia.dat; !name of data file
  type is individual; !data is individual observations
  format is free; !data is free of formats - space delimited

variable:
  names are mech vec algebra analysis stat; !variable names
  usevariables mech vec algebra analysis stat; !variables to use

analysis:
  type = efa 1 4; !perform efa and extract 1, 2, 3 and 4 factors
  rotation = cf-varimax (orthogonal); !choice of rotation
  estimator = ml; !maximum likelihood estimation
  iterations = 50; !maximum number of iterations

output:
  sampstat residual; !print sample statistics and residual correlation matrix
```

The code above has Mplus conduct an exploratory factor analysis. Since the commands (those in blue) have been explained above, here is a quick and dirty run down and explanation of the available options associated with the `analysis` and `output` commands. Since reading in data has been covered, the focus will be on the options for analysis:

(a) `type = efa n n`

Specifies that the type of modeling being fit to the data is an exploratory factor analysis. The first `n` relates to the smallest number of factors to be extracted. The second `n` defines the largest number of factors to extract. Mplus will output all solutions from smallest `n` to largest `n` factors extracted.

(b) `rotation = name(type)`

`name` specifies the family of rotations to be used and `type` relates to oblique or orthogonal. The default rotation is oblique oblimin rotation and this will be conducted by excluding this option. The various options available but not exhaustive are `quartimin(oblique)`, `promax(oblique)`, `cf-varimax(orthogonal)`, `cf-varimax(oblique)`, `geomin(orthogonal)` and `geomin(oblique)`. Note that for `promax` rotation, `kappa` is fixed at 3 in Mplus cannot be modified at present. Note that at present, Mplus does not have the capability of outputting unrotated factor loadings.

(c) `estimator = name`

Specifies the type of estimator used. Two known options are `ml = maximum likelihood` and `uls = unweighted least squares`.

(d) `iterations = n`

Specifies the maximum number of iterations to perform.

(e) `sampstat`

Tells Mplus to print out descriptive sample statistics such as the means and correlations of the sample data.

(f) `residual`

Tells Mplus to print out the residual correlation matrix.